
Pegasus is a PEG (Parsing Expression Grammar) parser generator for Elixir. It takes PEG grammar definitions and compiles them into efficient NimbleParsec parsers at compile time.

  
    
  
  Why Pegasus?


	Familiar syntax: Use standard PEG notation instead of learning NimbleParsec's combinator API
	Compile-time generation: Parsers are generated at compile time, with zero runtime overhead
	Full NimbleParsec power: Access all NimbleParsec features like post-traversal hooks, tagging, and tokenization
	Battle-tested format: PEG is a well-documented, widely-used grammar format


  
    
  
  Installation


Add pegasus to your list of dependencies in mix.exs:
def deps do
  [
    {:pegasus, "~> 1.0"}
  ]
end

  
    
  
  Quick Start


defmodule MyParser do
  require Pegasus

  # Define a simple parser for comma-separated numbers
  Pegasus.parser_from_string("""
    numbers <- number (',' number)*
    number  <- [0-9]+
  """,
    numbers: [parser: true],
    number: [collect: true]
  )
end

# Use the parser
MyParser.numbers("1,2,3")
# => {:ok, ["1", "2", "3"], "", %{}, {1, 0}, 5}

  
    
  
  How It Works


Pegasus operates in three stages:
	Parse: Your PEG grammar string is parsed into an AST
	Transform: The AST is converted into NimbleParsec combinators
	Compile: NimbleParsec compiles the combinators into an efficient parser

All of this happens at compile time via Elixir macros, so your final application contains optimized parsing code with no runtime grammar processing.

  
    
  
  PEG Grammar Syntax


Pegasus supports the standard PEG syntax:
	Syntax	Meaning
	'literal' or "literal"	Match exact string
	[a-z]	Character class (match one character in range)
	[^a-z]	Negated character class
	.	Match any single character
	e1 e2	Sequence (match e1 then e2)
	e1 / e2	Ordered choice (try e1, if it fails try e2)
	e*	Zero or more repetitions
	e+	One or more repetitions
	e?	Optional (zero or one)
	&e	Positive lookahead (match without consuming)
	!e	Negative lookahead (fail if matches)
	(e)	Grouping
	<e>	Extracted group (capture matched text)
	# comment	Line comment


  
    
  
  Example Grammar


# A simple arithmetic expression parser
expression <- term (('+' / '-') term)*
term       <- factor (('*' / '/') factor)*
factor     <- number / '(' expression ')'
number     <- [0-9]+

  
    
  
  Parser Options


Options are passed as a keyword list to configure how each rule is compiled:
Pegasus.parser_from_string(grammar,
  rule_name: [option: value, ...]
)

  
    
  
  Common Options


	Option	Description
	parser: true	Export as a parser function (entry point)
	export: true	Make combinator public instead of private
	collect: true	Merge matched content into a single binary
	token: :atom	Replace match with a token value
	tag: :atom	Wrap result in a tagged tuple {:atom, [...]}
	ignore: true	Discard the matched content
	post_traverse: {fun, args}	Apply a transformation function

See the Parser Options Guide for detailed documentation.

  
    
  
  Loading from Files


For larger grammars, load from a .peg file:
Pegasus.parser_from_file("priv/grammar.peg",
  start: [parser: true]
)

  
    
  
  Documentation


	Getting Started Guide - Tutorial introduction
	PEG Grammar Reference - Complete syntax reference
	Parser Options Guide - All configuration options
	Advanced Examples - Real-world patterns


  
    
  
  PEG Reference


For the original PEG specification, see:
https://www.piumarta.com/software/peg/peg.1.html

  
    
  
  License


MIT License - see LICENSE for details.


  

    
LICENSE
    


MIT License

Copyright (c) 2024 Isaac Yonemoto

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all
copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
SOFTWARE.



  

    
Getting Started with Pegasus
    

This guide walks you through creating your first parser with Pegasus.

  
    
  
  What is Pegasus?


Pegasus is a parser generator that lets you write parsers using PEG (Parsing Expression Grammar) notation. At compile time, Pegasus transforms your grammar into efficient NimbleParsec code.
If you're familiar with tools like YACC, Bison, or ANTLR, Pegasus serves a similar purpose but is designed specifically for Elixir and integrates seamlessly with NimbleParsec.

  
    
  
  Installation


Add Pegasus to your dependencies in mix.exs:
def deps do
  [
    {:pegasus, "~> 1.0"}
  ]
end
Then run:
mix deps.get


  
    
  
  Your First Parser


Let's build a simple parser that parses comma-separated integers.

  
    
  
  Step 1: Define the Grammar


Create a new module and define your grammar:
defmodule IntegerListParser do
  require Pegasus

  Pegasus.parser_from_string("""
    list   <- number (',' number)*
    number <- '-'? [0-9]+
  """,
    list: [parser: true],
    number: [collect: true]
  )
end
Let's break this down:
	require Pegasus - Required to use Pegasus macros
	parser_from_string/2 - The main macro that generates parsers
	The grammar defines two rules:	list matches a number followed by zero or more comma-separated numbers
	number matches an optional minus sign followed by one or more digits


	Options configure how rules are compiled:	parser: true makes list a callable entry point
	collect: true combines matched digits into a single string




  
    
  
  Step 2: Use the Parser


# Parse a valid list
IntegerListParser.list("1,2,3")
# => {:ok, ["1", "2", "3"], "", %{}, {1, 0}, 5}

# Parse with negative numbers
IntegerListParser.list("-5,10,-3")
# => {:ok, ["-5", "10", "-3"], "", %{}, {1, 0}, 9}

# Parse failure
IntegerListParser.list("not,numbers")
# => {:error, "expected ASCII character in the range \"0\" to \"9\"",
#     "not,numbers", %{}, {1, 0}, 0}

  
    
  
  Understanding the Result


A successful parse returns a 6-tuple:
{:ok, results, remaining, context, position, byte_offset}
	results - List of matched values
	remaining - Unparsed input (empty string if fully consumed)
	context - Parser context (starts as empty map)
	position - Current {line, line_offset} in the input
	byte_offset - Current byte position

A failed parse returns:
{:error, message, remaining, context, position, byte_offset}

  
    
  
  Converting Results


Raw string results aren't always useful. Use :post_traverse to transform them:
defmodule IntegerListParser do
  require Pegasus

  Pegasus.parser_from_string("""
    list   <- number (',' number)*
    number <- '-'? [0-9]+
  """,
    list: [parser: true],
    number: [collect: true, post_traverse: {:to_integer, []}]
  )

  defp to_integer(rest, [num_string], context, _position, _offset) do
    {rest, [String.to_integer(num_string)], context}
  end
end

IntegerListParser.list("1,2,3")
# => {:ok, [1, 2, 3], "", %{}, {1, 0}, 5}

  
    
  
  Adding Structure with Tags


For more complex parsers, you'll want structured output. Use :tag to wrap results:
defmodule KeyValueParser do
  require Pegasus

  Pegasus.parser_from_string("""
    pair  <- key '=' value
    key   <- [a-z]+
    value <- [0-9]+
  """,
    pair: [parser: true, tag: :pair],
    key: [collect: true, tag: :key],
    value: [collect: true, tag: :value]
  )
end

KeyValueParser.pair("foo=42")
# => {:ok, [{:pair, [{:key, "foo"}, {:value, "42"}]}], "", %{}, {1, 0}, 6}

  
    
  
  Ignoring Content


Use :ignore to discard matched content you don't need:
defmodule WordParser do
  require Pegasus

  Pegasus.parser_from_string("""
    words <- word (ws word)*
    word  <- [a-zA-Z]+
    ws    <- [ \\t]+
  """,
    words: [parser: true],
    word: [collect: true],
    ws: [ignore: true]
  )
end

WordParser.words("hello world")
# => {:ok, ["hello", "world"], "", %{}, {1, 0}, 11}
Without :ignore, the whitespace would appear in the results.

  
    
  
  Loading from Files


For larger grammars, store them in separate files:
# priv/grammar.peg
expression <- term (('+' / '-') term)*
term       <- number
number     <- [0-9]+
defmodule ExprParser do
  require Pegasus

  Pegasus.parser_from_file("priv/grammar.peg",
    expression: [parser: true],
    number: [collect: true]
  )
end

  
    
  
  Common Patterns



  
    
  
  Optional Whitespace


ws     <- [ \t\n\r]*
token  <- ws content ws

  
    
  
  Quoted Strings


string <- '"' (!'"' .)* '"'

  
    
  
  Comments


comment <- '#' (![\n\r] .)*

  
    
  
  Identifiers


ident <- [a-zA-Z_] [a-zA-Z0-9_]*

  
    
  
  Next Steps


	Read the PEG Grammar Reference for complete syntax documentation
	See Parser Options for all available configuration options
	Check out Advanced Examples for real-world parser patterns



  

    
PEG Grammar Reference
    

This guide provides a complete reference for the PEG (Parsing Expression Grammar) syntax supported by Pegasus.

  
    
  
  Grammar Structure


A PEG grammar consists of one or more definitions (rules). Each definition has an identifier and an expression:
identifier <- expression
The first rule is typically the entry point, though in Pegasus you explicitly mark entry points with the :parser option.

  
    
  
  Identifiers


Rule names (identifiers) must start with a letter or underscore, followed by letters, digits, or underscores:
my_rule    <- ...
Rule2      <- ...
_private   <- ...
Capitalized Identifiers
Capitalized identifiers like MyRule work but require special handling when
calling from Elixir code. See the Capitalized Identifiers section.


  
    
  
  Expressions



  
    
  
  Literals


Match exact strings using single or double quotes:
rule <- 'hello'
rule <- "world"
Both quote styles are equivalent. Use one when you need to include the other:
quoted <- "it's"
also   <- 'say "hello"'

  
    
  
  Character Classes


Match a single character from a set:
digit <- [0-9]
letter <- [a-zA-Z]
hex <- [0-9a-fA-F]
Combine multiple ranges and individual characters:
alphanum <- [a-zA-Z0-9_]
vowel <- [aeiouAEIOU]
Negated Classes
Match any character NOT in the set:
not_digit <- [^0-9]
not_quote <- [^"]
Special Characters in Classes
Use backslash escapes for special characters:
bracket <- [\[\]]        # matches [ or ]
hyphen <- [a\-z]         # matches a, -, or z (hyphen as literal)
backslash <- [\\]        # matches \

  
    
  
  Dot (Any Character)


Match any single character:
any <- .
This matches any character including newlines.

  
    
  
  Sequences


Match expressions in order:
hello_world <- 'hello' ' ' 'world'
All parts must match for the sequence to succeed.

  
    
  
  Ordered Choice


Try alternatives in order:
bool <- 'true' / 'false'
digit <- '0' / '1' / '2' / '3' / '4' / '5' / '6' / '7' / '8' / '9'
The first matching alternative wins. Unlike regular expressions, PEG choices are deterministic and ordered.

  
    
  
  Repetition


Zero or More (*)
digits <- [0-9]*
ws <- [ \t\n]*
One or More (+)
identifier <- [a-zA-Z] [a-zA-Z0-9]*
number <- [0-9]+
Optional (?)
signed_number <- '-'? [0-9]+
optional_semicolon <- ';'?

  
    
  
  Grouping


Use parentheses to group expressions:
term <- ('+' / '-') number
list <- item (',' item)*
Grouping is essential for combining operators:
# Without grouping: matches 'a' or ('b' followed by 'c')
wrong <- 'a' / 'b' 'c'

# With grouping: matches ('a' or 'b') followed by 'c'
right <- ('a' / 'b') 'c'

  
    
  
  Lookahead


Positive Lookahead (&)
Match only if the expression would match, but don't consume input:
# Match 'a' only if followed by 'b'
a_before_b <- 'a' &'b'
Negative Lookahead (!)
Match only if the expression would NOT match:
# Match any character except newline
not_newline <- !'\n' .

# Match identifier that isn't a keyword
identifier <- !keyword [a-zA-Z]+
keyword <- 'if' / 'else' / 'while'

  
    
  
  Extracted Groups (<...>)


Mark content for extraction:
quoted <- '"' <[^"]*> '"'
Extracted groups filter the result to include only the matched text, excluding surrounding syntax.

  
    
  
  Escape Sequences


Pegasus supports ANSI C escape sequences in literals and character classes:
	Escape	Meaning
	\a	Bell (alert)
	\b	Backspace
	\e	Escape
	\f	Form feed
	\n	Newline
	\r	Carriage return
	\t	Horizontal tab
	\v	Vertical tab
	\'	Single quote
	\"	Double quote
	\\	Backslash
	\[	Left bracket
	\]	Right bracket
	\-	Hyphen (in character classes)


  
    
  
  Octal Escapes


Specify characters by octal code:
null <- '\0'
bell <- '\7'
tab <- '\11'
max <- '\377'
Octal escapes use 1-3 digits, with values from 0-377 (octal).

  
    
  
  Comments


Line comments start with #:
# This is a comment
rule <- 'hello'  # inline comment

  
    
  
  Operator Precedence


From highest to lowest precedence:
	() - Grouping
	*, +, ? - Repetition
	&, ! - Lookahead
	Sequence (implicit)
	/ - Choice

Example:
# This parses as: (a b*) / c
rule <- a b* / c

# Use grouping to change precedence:
rule <- a (b / c)*
Capitalized Identifiers
Capitalized PEG identifiers like Statement or Expression work fine.
Just remember to put a colon in front of them in the options keyword list,
since capitalized names in Elixir are aliases:
Pegasus.parser_from_string("Foo <- 'foo'", Foo: [parser: :parse])
Capitalized identifiers also require special handling when called directly.
You can wrap in a lowercase combinator or use apply/3:
defparsec :parse, parsec(:Foo)
# or
apply(MyParser, :Foo, ["foo"])


  
    
  
  Common Patterns



  
    
  
  Whitespace Handling


ws <- [ \t\n\r]*
token <- ws content ws

  
    
  
  Quoted Strings


string <- '"' (!'"' .)* '"'
With escape sequences:
string <- '"' (escape / !'"' .)* '"'
escape <- '\\' [nrt"\\]

  
    
  
  Comments (C-style)


line_comment <- '//' (!'\n' .)* '\n'
block_comment <- '/*' (!'*/' .)* '*/'

  
    
  
  Identifiers


identifier <- [a-zA-Z_] [a-zA-Z0-9_]*

  
    
  
  Numbers


integer <- '-'? [0-9]+
float <- '-'? [0-9]+ '.' [0-9]+

  
    
  
  Separated Lists


# Comma-separated with optional trailing comma
list <- item (',' item)* ','?

# At least one item
nonempty_list <- item (',' item)*

  
    
  
  Debugging Tips


	Start simple: Build your grammar incrementally, testing each rule.

	Use :parser: Mark rules you want to test with parser: true so you can call them directly.

	Check precedence: When in doubt, add parentheses to make grouping explicit.

	Ordered choice matters: Put more specific alternatives first:
# Wrong: 'if' matches before 'ifelse'
keyword <- 'if' / 'ifelse'

# Right: longer match first
keyword <- 'ifelse' / 'if'

	Avoid left recursion: PEG parsers don't support left recursion.


Left Recursion
Left-recursive rules will cause infinite loops:
# This will infinite loop!
expr <- expr '+' term

# Use iteration instead
expr <- term ('+' term)*



  

    
Parser Options Reference
    

This guide documents all options available when configuring Pegasus parsers.

  
    
  
  Overview


Options are passed as a keyword list to parser_from_string/2 or parser_from_file/2. Each key is a rule name, and each value is a list of options for that rule:
Pegasus.parser_from_string("""
  start <- number (',' number)*
  number <- [0-9]+
""",
  start: [parser: true],
  number: [collect: true, tag: :num]
)

  
    
  
  Option Application Order


Options are applied in this order:
	:start_position - Inject position information
	:collect - Merge content into binary
	:token - Replace with token value
	:tag - Wrap in tagged tuple
	:post_traverse - Custom transformation
	:ignore - Discard content
	:alias - Substitute combinator

This order matters! For example, :collect happens before :tag, so you can collect characters into a string and then tag the result.

  
    
  
  Options Reference



  
    
  
  :parser


Marks a rule as an entry point that can be called directly.
Values:
	true - Create a parser with the rule name
	:custom_name - Create a parser with a custom name

Without :parser: Rules become private combinators only usable by other rules.
With :parser: The rule becomes a callable function returning {:ok, result, rest, ...} or {:error, ...}.
Pegasus.parser_from_string("""
  start <- greeting name
  greeting <- 'Hello, '
  name <- [a-zA-Z]+
""", start: [parser: true])

# Now you can call:
MyModule.start("Hello, World")
Renaming:
Pegasus.parser_from_string("""
  internal_name <- [a-z]+
""", internal_name: [parser: :parse])

# Creates `parse/1` instead of `internal_name/1`
MyModule.parse("hello")

  
    
  
  :export


Makes a combinator public instead of private.
Values:
	true - Export the combinator

Default: Combinators are private (defcombinatorp).
Use this when you need to reference a combinator from outside the module or compose it with other NimbleParsec combinators.
Pegasus.parser_from_string("""
  number <- [0-9]+
""", number: [export: true, collect: true])

# In another module:
import NimbleParsec
defparsec :my_parser, parsec({OtherModule, :number})

  
    
  
  :collect


Merges all matched content into a single binary string.
Values:
	true - Collect into binary

Without :collect, character matches remain as separate elements:
# Without collect
# [0-9]+ matching "123" produces: [49, 50, 51] (character codes)

# With collect
# [0-9]+ matching "123" produces: "123"
Collect Requirements
When using :collect, all content must be iodata (binaries or character codes).
Tags and tokens will cause errors.

Pegasus.parser_from_string("""
  number <- sign? digits
  sign <- '-' / '+'
  digits <- [0-9]+
""",
  number: [parser: true, collect: true]
)

MyModule.number("-42")
# => {:ok, ["-42"], "", ...}

  
    
  
  :token


Replaces matched content with a token value.
Values:
	true - Use the rule name as the token
	:custom - Use a custom atom as the token
	Any term - Use that value as the token

Tokens are useful for lexer-style parsing where you want to classify matches:
Pegasus.parser_from_string("""
  keyword <- 'if' / 'else' / 'while' / 'for'
""", keyword: [parser: true, collect: true, token: :keyword])

MyModule.keyword("while")
# => {:ok, [:keyword], "", ...}
With token: true:
Pegasus.parser_from_string("""
  if_keyword <- 'if'
  else_keyword <- 'else'
""",
  if_keyword: [token: true],
  else_keyword: [token: true]
)

# 'if' becomes :if_keyword
# 'else' becomes :else_keyword

  
    
  
  :tag


Wraps the result in a tagged tuple {tag, content}.
Values:
	true - Use the rule name as the tag
	:custom - Use a custom atom as the tag
	Any term - Use that value as the tag

Pegasus.parser_from_string("""
  pair <- key '=' value
  key <- [a-z]+
  value <- [0-9]+
""",
  pair: [parser: true],
  key: [collect: true, tag: :key],
  value: [collect: true, tag: :value]
)

MyModule.pair("foo=42")
# => {:ok, [{:key, "foo"}, {:value, "42"}], "", ...}
Nested tags:
Pegasus.parser_from_string("""
  pair <- key '=' value
  key <- [a-z]+
  value <- [0-9]+
""",
  pair: [parser: true, tag: :pair],
  key: [collect: true, tag: :key],
  value: [collect: true, tag: :value]
)

MyModule.pair("foo=42")
# => {:ok, [{:pair, [{:key, "foo"}, {:value, "42"}]}], "", ...}

  
    
  
  :ignore


Discards matched content from the result.
Values:
	true - Ignore the matched content

Use this for whitespace, delimiters, and other syntax that shouldn't appear in the result:
Pegasus.parser_from_string("""
  list <- item (sep item)*
  item <- [a-z]+
  sep <- [ ,]+
""",
  list: [parser: true],
  item: [collect: true],
  sep: [ignore: true]
)

MyModule.list("foo, bar, baz")
# => {:ok, ["foo", "bar", "baz"], "", ...}
# Without ignore: ["foo", ", ", "bar", ", ", "baz"]

  
    
  
  :start_position


Injects position information at the start of the match.
Values:
	true - Add position map

The position map contains:
	:line - Line number (1-based)
	:column - Column number (1-based)
	:offset - Byte offset from start

Pegasus.parser_from_string("""
  token <- [a-z]+
""", token: [parser: true, start_position: true, collect: true])

MyModule.token("hello")
# => {:ok, [%{line: 1, column: 1, offset: 0}, "hello"], "", ...}
This is useful for error reporting and source mapping.

  
    
  
  :post_traverse


Applies a custom transformation function after parsing.
Values:
	{:function_name, args} - Call function with additional args
	:function_name - Shorthand for {:function_name, []}

The function receives:
	rest - Remaining unparsed input
	args - Matched content (in reversed order!)
	context - Parser context
	position - {line, line_offset} tuple
	byte_offset - Current byte position

It must return {rest, new_args, context}.
defmodule MyParser do
  require Pegasus

  Pegasus.parser_from_string("""
    number <- '-'? [0-9]+
  """, number: [parser: true, collect: true, post_traverse: {:to_integer, []}])

  defp to_integer(rest, [num_string], context, _position, _offset) do
    {rest, [String.to_integer(num_string)], context}
  end
end

MyParser.number("-42")
# => {:ok, [-42], "", ...}
Arguments Are Reversed
The args parameter is in reversed order from how content was matched.

Pegasus.parser_from_string("""
  pair <- first ',' second
  first <- [a-z]+
  second <- [0-9]+
""", pair: [parser: true, post_traverse: {:handle_pair, []}])

# When parsing "abc,123":
defp handle_pair(rest, [second, ",", first], context, _, _) do
  # second = "123", first = "abc" (reversed!)
  {rest, [{first, second}], context}
end
With extra arguments:
Pegasus.parser_from_string("""
  number <- [0-9]+
""", number: [collect: true, post_traverse: {:parse_base, [16]}])

defp parse_base(rest, [digits], context, _, _, base) do
  {rest, [String.to_integer(digits, base)], context}
end

  
    
  
  :alias


Substitutes a custom combinator in place of the grammar rule.
Values:
	:combinator_name - Use this combinator instead

The grammar rule is ignored; only the referenced combinator is used.
defmodule MyParser do
  require Pegasus
  import NimbleParsec

  # Define a custom combinator
  defcombinatorp :my_special_parser,
    ascii_string([?a..?z], min: 1)
    |> map({String, :upcase, []})

  Pegasus.parser_from_string("""
    word <- [a-z]+
  """, word: [parser: true, alias: :my_special_parser])
end

MyParser.word("hello")
# => {:ok, ["HELLO"], "", ...}
Use :alias when:
	The PEG syntax can't express what you need
	You need NimbleParsec-specific features
	You're gradually migrating from NimbleParsec to Pegasus


  
    
  
  Combining Options


Options compose naturally:
Pegasus.parser_from_string("""
  token <- [a-zA-Z]+
""",
  token: [
    parser: true,
    start_position: true,  # 1. Add position
    collect: true,         # 2. Merge to string
    tag: :identifier       # 3. Wrap in tuple
  ]
)

MyModule.token("hello")
# => {:ok, [{:identifier, [%{line: 1, column: 1, offset: 0}, "hello"]}], "", ...}

  
    
  
  Common Patterns



  
    
  
  Lexer Tokens


Pegasus.parser_from_string("""
  tokens <- (keyword / identifier / number / ws)*
  keyword <- 'if' / 'else' / 'while'
  identifier <- [a-zA-Z_] [a-zA-Z0-9_]*
  number <- [0-9]+
  ws <- [ \\t\\n]+
""",
  tokens: [parser: true],
  keyword: [collect: true, token: :keyword],
  identifier: [collect: true, token: :ident],
  number: [collect: true, token: :number],
  ws: [ignore: true]
)

  
    
  
  AST Building


Pegasus.parser_from_string("""
  expr <- term (('+' / '-') term)*
  term <- number
  number <- [0-9]+
""",
  expr: [parser: true, tag: :expr],
  term: [tag: :term],
  number: [collect: true, post_traverse: {:to_int, []}]
)

  
    
  
  Error Positions


Pegasus.parser_from_string("""
  statement <- keyword ws expr
  keyword <- 'let' / 'var'
  expr <- [a-z]+
  ws <- [ ]+
""",
  statement: [parser: true],
  keyword: [start_position: true, collect: true, tag: :keyword],
  expr: [start_position: true, collect: true, tag: :expr],
  ws: [ignore: true]
)


  

    
Advanced Examples
    

This guide demonstrates advanced Pegasus patterns from real-world parsers.

  
    
  
  Vendored Grammar Files


Use parser_from_file/2 when you have a reference PEG grammar vendored into your codebase. This is common when implementing a parser for an established format or language that has an official PEG specification:
defmodule MyParser do
  require Pegasus

  # Grammar vendored from an upstream specification
  @external_resource Path.join(__DIR__, "grammar/grammar.peg")

  Pegasus.parser_from_file(
    Path.join(__DIR__, "grammar/grammar.peg"),
    @parser_options
  )
end
Using @external_resource ensures the module recompiles when the grammar file changes.
This approach keeps the authoritative grammar separate from your Elixir code, making it easier to update when the upstream specification changes.

  
    
  
  Dynamic Option Generation


When you have many similar rules, generate options programmatically:
defmodule MyParser do
  require Pegasus

  # Keywords that become tokens
  @keywords ~w[if else while for return]a
  @keyword_options Enum.map(@keywords, fn kw ->
    {:"KEYWORD_#{kw}", [token: kw]}
  end)

  # Operators with position tracking
  @operators %{PLUS: :+, MINUS: :-, STAR: :*, SLASH: :/}
  @operator_options Enum.map(@operators, fn {name, op} ->
    {name, [token: op, start_position: true]}
  end)

  # Collected tokens
  @collected ~w[INTEGER FLOAT IDENTIFIER]a
  @collected_options Enum.map(@collected, fn name ->
    {name, [collect: true, post_traverse: {:parse_token, [name]}]}
  end)

  @parser_options [
    root: [parser: true]
  ] ++ @keyword_options ++ @operator_options ++ @collected_options

  Pegasus.parser_from_string(@grammar, @parser_options)
end

  
    
  
  Building AST Nodes with Structs


Define structs for your AST nodes:
defmodule MyParser.Function do
  defstruct [
    :name,
    :params,
    :return_type,
    :body,
    :location,
    public: false,
    inline: false
  ]
end
Use post-traverse to build them:
defmodule MyParser do
  require Pegasus
  alias MyParser.Function

  Pegasus.parser_from_string("""
    function <- 'pub'? 'inline'? 'fn' identifier params return_type? block
    identifier <- [a-zA-Z_] [a-zA-Z0-9_]*
    params <- '(' param_list? ')'
    param_list <- param (',' param)*
    param <- identifier ':' type
    return_type <- '->' type
    type <- identifier
    block <- '{' statement* '}'
    statement <- [^}]*
  """,
    function: [parser: true, tag: true, start_position: true,
               post_traverse: {Function, :post_traverse, []}],
    identifier: [collect: true],
    params: [tag: true],
    return_type: [tag: true],
    block: [tag: true]
  )
end

defmodule MyParser.Function do
  # ... struct definition ...

  def post_traverse(rest, [{:function, [start | args]} | rest_args], context, _, _) do
    func = parse(args, %__MODULE__{})
    func = %{func | location: {start.line, start.column}}
    {rest, [func | rest_args], context}
  end

  defp parse([:pub | rest], func), do: parse(rest, %{func | public: true})
  defp parse([:inline | rest], func), do: parse(rest, %{func | inline: true})
  defp parse([:fn, name | rest], func), do: parse(rest, %{func | name: name})
  defp parse([{:params, params} | rest], func), do: parse(rest, %{func | params: params})
  defp parse([{:return_type, [:"->", type]} | rest], func), do: parse(rest, %{func | return_type: type})
  defp parse([{:block, body}], func), do: %{func | body: body}
  defp parse([], func), do: func
end

  
    
  
  Operator Precedence with Shunting-Yard


PEG grammars don't directly express operator precedence. Implement it with post-traverse:
defmodule ExprParser do
  require Pegasus

  Pegasus.parser_from_string("""
    expr <- term (operator term)*
    term <- number / '(' expr ')'
    operator <- '+' / '-' / '*' / '/'
    number <- [0-9]+
  """,
    expr: [parser: true, tag: :Expr, post_traverse: {__MODULE__, :build_tree, []}],
    operator: [collect: true],
    number: [collect: true, post_traverse: {:to_integer, []}]
  )

  defp to_integer(rest, [num], context, _, _) do
    {rest, [String.to_integer(num)], context}
  end

  # Operator precedence (higher = binds tighter)
  @precedence %{"*" => 2, "/" => 2, "+" => 1, "-" => 1}

  def build_tree(rest, [{:Expr, args} | rest_args], context, _, _) do
    tree = shunting_yard(args, [], [])
    {rest, [tree | rest_args], context}
  end

  # Shunting-yard algorithm
  defp shunting_yard([], [], [result]), do: result
  defp shunting_yard([], [op | ops], output) do
    shunting_yard([], ops, apply_op(op, output))
  end

  defp shunting_yard([term | rest], ops, output) when is_integer(term) do
    shunting_yard(rest, ops, [term | output])
  end

  defp shunting_yard([op | rest], [], output) do
    shunting_yard(rest, [op], output)
  end

  defp shunting_yard([op | rest], [top | ops], output) do
    if @precedence[top] >= @precedence[op] do
      shunting_yard([op | rest], ops, apply_op(top, output))
    else
      shunting_yard(rest, [op, top | ops], output)
    end
  end

  defp apply_op(op, [b, a | rest]) do
    [{String.to_atom(op), a, b} | rest]
  end
end

ExprParser.expr("1+2*3")
# => {:ok, [{:+, 1, {:*, 2, 3}}], "", ...}

  
    
  
  Context for Accumulating State


Use the parser context to collect information across the entire parse:
defmodule DocParser do
  require Pegasus

  defstruct comments: [], imports: []

  Pegasus.parser_from_string("""
    root <- (comment / import / statement)*
    comment <- '//' (!'\n' .)*
    import <- 'import' ws path
    path <- [a-zA-Z/]+
    statement <- (!'\n' .)* '\n'
    ws <- [ \t]+
  """,
    root: [parser: :parse],
    comment: [tag: true, post_traverse: {:collect_comment, []}],
    import: [tag: true, post_traverse: {:collect_import, []}],
    path: [collect: true],
    ws: [ignore: true],
    statement: [ignore: true]
  )

  # Initialize context
  def parse(input) do
    case parser(input, context: %__MODULE__{}) do
      {:ok, _, "", context, _, _} ->
        {:ok, %{context | comments: Enum.reverse(context.comments),
                         imports: Enum.reverse(context.imports)}}
      error -> error
    end
  end

  defp collect_comment(rest, [{:comment, chars} | args], context, _, _) do
    comment = chars |> tl() |> tl() |> List.to_string()
    {rest, args, %{context | comments: [comment | context.comments]}}
  end

  defp collect_import(rest, [{:import, [:import, path]} | args], context, _, _) do
    {rest, args, %{context | imports: [path | context.imports]}}
  end
end

  
    
  
  Location Tracking for Error Messages


Track source locations for better error reporting:
defmodule LocatedParser do
  require Pegasus

  defmodule Token do
    defstruct [:type, :value, :line, :column]
  end

  Pegasus.parser_from_string("""
    tokens <- token*
    token <- keyword / identifier / number / ws
    keyword <- 'if' / 'else' / 'while'
    identifier <- [a-zA-Z_] [a-zA-Z0-9_]*
    number <- [0-9]+
    ws <- [ \t\n]+
  """,
    tokens: [parser: true],
    keyword: [start_position: true, collect: true,
              post_traverse: {:make_token, [:keyword]}],
    identifier: [start_position: true, collect: true,
                 post_traverse: {:make_token, [:identifier]}],
    number: [start_position: true, collect: true,
             post_traverse: {:make_token, [:number]}],
    ws: [ignore: true]
  )

  defp make_token(rest, [value, pos | args], context, _, _, type) do
    token = %Token{
      type: type,
      value: value,
      line: pos.line,
      column: pos.column
    }
    {rest, [token | args], context}
  end
end

  
    
  
  Combining with Custom NimbleParsec Combinators


Use :alias to integrate custom combinators:
defmodule UnicodeParser do
  require Pegasus
  import NimbleParsec

  # Custom combinator for Unicode categories
  defcombinatorp :unicode_letter,
    utf8_char([?a..?z, ?A..?Z, 0x00C0..0x00FF, 0x0100..0x017F])

  defcombinatorp :unicode_digit,
    utf8_char([?0..?9, 0x0660..0x0669, 0x06F0..0x06F9])

  Pegasus.parser_from_string("""
    identifier <- letter (letter / digit)*
    letter <- [a-zA-Z]
    digit <- [0-9]
  """,
    identifier: [parser: true, collect: true],
    letter: [alias: :unicode_letter],
    digit: [alias: :unicode_digit]
  )
end

  
    
  
  SQL Parser Example


A complete example showing tagged rules, post-traverse for AST building, and ignored tokens:
defmodule SQLParser do
  require Pegasus

  @options [
    # Entry point
    statement: [parser: true],

    # AST nodes with post-traverse
    select: [tag: "select", post_traverse: {:build_select, []}],
    where: [tag: "where"],
    from: [tag: "from"],

    # Binary expressions
    binary_expr: [tag: "binary", post_traverse: {:build_binary, []}],

    # Terminals
    identifier: [collect: true, tag: "identifier"],
    integer: [collect: true, post_traverse: {:to_int, []}],
    string: [collect: true],

    # Operators become tokens
    eq: [token: :=],
    and_op: [token: :and],
    or_op: [token: :or],

    # Ignored syntax
    ws: [ignore: true],
    comma: [ignore: true],
    semicolon: [ignore: true]
  ]

  Pegasus.parser_from_string("""
    statement <- ws select ws semicolon ws

    select <- 'SELECT' ws select_list ws from ws where?

    select_list <- '*' / expr_list
    expr_list <- expr (ws comma ws expr)*

    from <- 'FROM' ws identifier

    where <- 'WHERE' ws expr

    expr <- binary_expr / term
    binary_expr <- term ws operator ws expr
    term <- identifier / integer / string

    operator <- eq / and_op / or_op
    eq <- '='
    and_op <- 'AND'
    or_op <- 'OR'

    identifier <- [a-zA-Z_] [a-zA-Z0-9_]*
    integer <- [0-9]+
    string <- ['] (!['] .)* [']

    ws <- [ \\t\\n]*
    comma <- ','
    semicolon <- ';'
  """, @options)

  defp to_int(rest, [num], context, _, _) do
    {rest, [String.to_integer(num)], context}
  end

  defp build_select(rest, [{_, args} | rest_args], context, _, _) do
    ast = %{
      type: :select,
      columns: extract_columns(args),
      from: extract_from(args),
      where: extract_where(args)
    }
    {rest, [ast | rest_args], context}
  end

  defp build_binary(rest, [{_, [left, op, right]} | rest_args], context, _, _) do
    ast = %{type: :binary, operator: op, left: left, right: right}
    {rest, [ast | rest_args], context}
  end

  # Helper functions...
  defp extract_columns(args), do: # ...
  defp extract_from(args), do: # ...
  defp extract_where(args), do: # ...
end

  
    
  
  Modular Parser Organization


For large parsers, organize into modules by grammar section:
lib/
  my_parser.ex           # Main module with parser_from_file
  my_parser/
    expression.ex        # Expression AST and post_traverse
    statement.ex         # Statement AST and post_traverse
    function.ex          # Function AST and post_traverse
    types.ex             # Type AST and post_traverse
    collected.ex         # Token collection handlers
grammar/
  grammar.peg            # The PEG grammar
Each module handles its own AST nodes:
# lib/my_parser/expression.ex
defmodule MyParser.Expression do
  defstruct [:operator, :left, :right, :location]

  def post_traverse(rest, [{:Expr, args} | rest_args], context, _, _) do
    expr = build(args)
    {rest, [expr | rest_args], context}
  end

  defp build(args) do
    # ...
  end
end
And the main module references them:
# lib/my_parser.ex
defmodule MyParser do
  require Pegasus
  alias MyParser.Expression
  alias MyParser.Statement
  alias MyParser.Function

  @parser_options [
    expr: [tag: :Expr, post_traverse: {Expression, :post_traverse, []}],
    statement: [tag: :Statement, post_traverse: {Statement, :post_traverse, []}],
    function: [tag: :Function, post_traverse: {Function, :post_traverse, []}]
  ]

  Pegasus.parser_from_file("grammar/grammar.peg", @parser_options)
end

  
    
  
  Tips for Large Parsers


	Start with the grammar: Get the PEG grammar working before adding options
	Add options incrementally: Start with parser: true on the entry point, then add more
	Test post-traverse in isolation: Write unit tests for transformation functions
	Use :tag liberally: Tagged output makes pattern matching in post-traverse easier
	Accumulate in reverse: Prepend to lists in post-traverse, reverse at the end
	Handle all variants: Use multiple function clauses to match different parse shapes
	Track locations early: Add start_position: true before you need it

A PEG (Parsing Expression Grammar) parser generator for Elixir.
Pegasus compiles PEG grammar definitions into efficient NimbleParsec parsers
at compile time. This gives you the familiar, readable PEG syntax while leveraging
NimbleParsec's optimized parsing engine.

  
    
  
  Quick Start


defmodule MyParser do
  require Pegasus

  Pegasus.parser_from_string("""
    numbers <- number (',' number)*
    number  <- [0-9]+
  """,
    numbers: [parser: true],
    number: [collect: true]
  )
end

MyParser.numbers("1,2,3")
# => {:ok, ["1", "2", "3"], "", %{}, {1, 0}, 5}

  
    
  
  Main API


	parser_from_string/2 - Define parsers from a PEG grammar string
	parser_from_file/2 - Load and compile a PEG grammar from a file
	parser_from_ast/2 - Advanced: compile a pre-parsed AST


  
    
  
  PEG Grammar Syntax


Pegasus supports the standard PEG syntax. For the full specification, see the
PEG reference.

  
    
  
  Rules


Rules are defined with the <- operator:
identifier <- expression

  
    
  
  Expressions


	Syntax	Description	Example
	'...' or "..."	Literal string	'hello'
	[...]	Character class	[a-zA-Z]
	[^...]	Negated character class	[^0-9]
	.	Any character	.
	e1 e2	Sequence	'a' 'b'
	e1 / e2	Ordered choice	'a' / 'b'
	e*	Zero or more	[0-9]*
	e+	One or more	[0-9]+
	e?	Optional	'-'?
	&e	Positive lookahead	&'x'
	!e	Negative lookahead	!'x'
	(e)	Grouping	('a' 'b')*
	<e>	Extracted group	<[a-z]+>
	# ...	Comment	# this is ignored


  
    
  
  Escape Sequences


Pegasus supports ANSI C escape sequences in literals and character classes:
	\a - bell
	\b - backspace
	\e - escape
	\f - form feed
	\n - newline
	\r - carriage return
	\t - tab
	\v - vertical tab
	\' - single quote
	\" - double quote
	\\ - backslash
	\[ and \] - brackets (useful in character classes)
	\- - literal hyphen (in character classes)
	\377 - octal escape (1-3 octal digits)


  
    
  
  Parser Options


Options control how each grammar rule is compiled. Pass them as a keyword list
where keys are rule names:
Pegasus.parser_from_string(grammar,
  rule_name: [option: value, ...]
)
Options are applied in the order specified.

  
    
  
  :parser


Export the rule as a parser function (an entry point that can be called directly).
Without this option, rules become private combinators.
Pegasus.parser_from_string("""
  start <- greeting name
  greeting <- 'Hello, '
  name <- [a-zA-Z]+
""", start: [parser: true])
The :parser option also accepts an atom to rename the parser:
Pegasus.parser_from_string("foo <- 'foo'", foo: [parser: :parse])
# Creates `parse/1` instead of `foo/1`

  
    
  
  :export


Make a combinator public instead of private. Use this when you need to reference
the combinator from other modules or compose it with other NimbleParsec combinators.
Pegasus.parser_from_string("""
  foo <- 'foo'
""", foo: [export: true])

  
    
  
  :collect


Merge all matched content into a single binary string. Useful for rules that match
multiple characters you want combined.
Pegasus.parser_from_string("""
  number <- [0-9]+
""", number: [collect: true])

# Without collect: ["1", "2", "3"]
# With collect: "123"
Collect requirements
When using :collect, all nested combinators must leave only iodata
(binaries/charlists) in the result. Tags and tokens will cause errors.


  
    
  
  :token


Replace the matched content with a token value.
	token: true - Use the rule name as the token

	token: :custom - Use a custom atom as the token
  Pegasus.parser_from_string("""
operator <- '+' / '-' / '*' / '/'
  """, operator: [collect: true, token: :op])
  # Matched "+" becomes :op



  
    
  
  :tag


Wrap the result in a tagged tuple {tag, content}.
	tag: true - Use the rule name as the tag

	tag: :custom - Use a custom atom as the tag
  Pegasus.parser_from_string("""
number <- [0-9]+
  """, number: [collect: true, tag: :num])
  # Result: {:num, "123"}



  
    
  
  :ignore


Discard the matched content. Useful for whitespace and delimiters.
Pegasus.parser_from_string("""
  list <- item (',' item)*
  item <- [a-z]+
""",
  list: [parser: true],
  item: [collect: true]
)

  
    
  
  :start_position


Inject position information at the start of the match. Adds a map with
:line, :column, and :offset keys.
Pegasus.parser_from_string("""
  token <- [a-z]+
""", token: [start_position: true, collect: true])

# Result: [%{line: 1, column: 0, offset: 0}, "hello"]

  
    
  
  :post_traverse


Apply a custom transformation function after the rule matches. The function
receives the parsing state and can transform the results.
Pegasus.parser_from_string("""
  number <- [0-9]+
""", number: [collect: true, post_traverse: {:to_integer, []}])

defp to_integer(rest, [num_string], context, _position, _offset) do
  {rest, [String.to_integer(num_string)], context}
end
Arguments are reversed
The second argument (matched content) is in reversed order from how
it was matched. Plan accordingly when pattern matching.


  
    
  
  :alias


Substitute a custom combinator in place of the grammar rule. Useful when
you need special handling that PEG syntax can't express.
Pegasus.parser_from_string("""
  special <- 'x'
""", special: [alias: :my_custom_combinator])
You must define my_custom_combinator as a NimbleParsec combinator in
your module.
Capitalized Identifiers
Capitalized PEG identifiers like Statement or Expression work fine.
Just remember to put a colon in front of them in the options keyword list,
since capitalized names in Elixir are aliases:
Pegasus.parser_from_string("Foo <- 'foo'", Foo: [parser: :parse])
Capitalized identifiers also require special handling when called directly.
You can wrap in a lowercase combinator or use apply/3:
defparsec :parse, parsec(:Foo)
# or
apply(MyParser, :Foo, ["foo"])


  
    
  
  Loading from Files


For larger grammars, store them in .peg files:
# In lib/my_parser.ex
Pegasus.parser_from_file("priv/grammar.peg",
  start: [parser: true]
)

  
    
  
  Output Format


Parsers return the standard NimbleParsec result tuple:
{:ok, results, remaining, context, position, byte_offset}
Or on failure:
{:error, message, remaining, context, position, byte_offset}
See NimbleParsec documentation for details.
Not Implemented
PEG actions (C code blocks like { code }) are not supported, as they are
specific to the C implementation. Use :post_traverse for custom transformations.


      


      
        Summary


  
    Functions
  


    
      
        OEBPS/dist/epub-CB7BJMUW.js
